Maximo Business Rules scripting
Maximo® Business Rules (MBR) is a scripting language that provides a cloud-safe way to extend
application business logic. MBR, JavaScript and Python are the scripting languages that are available in
IBM® Maximo EAM SaaS Flex. MBR is the only scripting language that is available in IBM Maximo EAM
SaaS.
A prerequisite for creating scripts using MBR is a full understanding of the principles and practices of
automation scripting.
MBR does not support input-output interaction.
This document provides information on the following aspects of MBR:
•
•
•
•
•

Language fundamentals
Supported operators
Supported functions
Supported object attribute access notations
Script code examples

Language fundamentals
MBR scripts are text files that are made up of expressions. These expressions contain operators,
functions, and variables.
When you create an MBR script, you must associate a launch point with the script. The launch point
specifies the context that the script runs in. For example, you can configure a script that runs when an
attribute is updated. MBR supports the definition of script launch points for the following Maximo
artifacts:
•
•
•
•

Maximo business objects (MBOs)
Attributes
Conditions
Actions

In MBR script code, you can perform the following tasks:
•
•
•

Define a function. You must prefix the function name with :.
Create a library script for reusable code. You can invoke a library script from any other script.
Add a comment. You must prefix the comment with #.

In MBR script code, you cannot perform the following tasks:
•
•
•

Access Java classes directly.
Call functions or library scripts recursively.
Define more than one expression in the same line. Expressions are newline delimited.

•
•

Create an expression that spans multiple lines.
Add inline comments. For example, the following inline comment is not allowed:
newmbo("opressure","assetmeter") #this is not allowed – put the comment in a new line

In addition, MBR restricts looping capability to the following items:
•
•

Delimited strings, also known as tokens.
MBOs that are related to the MBO that owns the current artifact. This set of related MBOs is
known as an MBOSet.

MBR script code is syntax-validated when you save your script.

Supported operators
Mathematical operators
Operator
+
*
/
%
^

Description
Addition or concatenation operator
Subtraction operator
Multiplication operator
Division operator
Modulo operator
Power operator

Boolean operators
Operator
=
==
!=
<>
<
<=
>
>=
&&
||

Description
Equal
Equal
Not equal
Not equal
Less than
Less than or equal to
Greater than
Greater than or equal to
Boolean AND
Boolean OR

Note that the Boolean NOT operation is implemented by the not(expression) function instead of the
traditional ! operator.

Supported Functions

Common mathematical functions
Function
not(expr)
random()
min(expr1,expr2)
max(expr1,expr2)
abs(expr)
round(expr)
floor(expr)
ceiling (expr)
log(expr)
log10(expr)
sqrt(expr)
sin(expr)
cos(expr)
tan(expr)
asin(expr)
acos(expr)
atan(expr)
sinh(expr)
cosh(expr)
tanh(expr)
rad(expr)

deg(expr)

pct(expr1,expr2)

Description
Boolean negation. Returns 1 if the expression is
not 0.
Produces a random number between 0 and 1.
Returns the smaller of two expressions.
Returns the bigger of two expressions.
Returns the absolute value of an expression.
Rounds a value by using the current rounding
mode.
Rounds a value down to the nearest integer.
Rounds a value up to the nearest integer.
Returns the natural logarithm of an expression.
Returns the common logarithm of an expression.
Returns the square root of an expression.
Returns the trigonometric sine of an angle in
degrees.
Returns the trigonometric cosine of an angle in
degrees.
Returns the trigonometric tangent of an angle in
degrees.
Returns the inverse trigonometric sine of an
angle in degrees.
Returns the inverse trigonometric cosine of an
angle in degrees.
Returns the inverse trigonometric tangent of an
angle in degrees.
Returns the hyperbolic sine of an angle.
Returns the hyperbolic cosine of an angle.
Returns the hyperbolic tangent of an angle.
Converts an angle that is measured in degrees to
an approximately equivalent angle that is
measured in radians.
Converts an angle that is measured in radians to
an approximately equivalent angle that is
measured in degrees.
Returns the equivalent percentage value of expr1
divided by expr2.

Variable setting and variable retrieval functions
Function
setvar(varname, expr[, global])

Description
Sets or creates the varname variable to have a
value of expr. The scope of the variable is local by

getvar(varname)

default. You can set the global parameter to
TRUE to make the scope of the variable global.
Retrieves the value of the varname variable. If
the variable is defined to have both local and
global scopes, the function returns the local
value.

Maximo business-specific functions
Function
invokescript(scriptname|functionname)

newmbo(varname, relation[, global])

setvaluetombo(varname, attrname, attrval)

setvalue(attrname, attrval)

setvaluenull(attrname)
setvaluenulltombo(varname, attrname)

error(errgrp, errkey)
setrequired(attrname, bool)

setreadonly(attrname, bool)

sethidden(attrname, bool)
deletethismbo(varname)
deleteall(relation)
deleteallfromthismbo(varname, relation)

Description
Invokes either a library script or a function that is
local to the current script. If you invoke a local
function, you must prefix the function name with
:.
Adds a new MBO to the MBOSet for the relation
relationship. The newly created MBO is visible in
the varname variable. The default scope of the
variable is local. You can set the global parameter
to TRUE to make the scope of the variable global.
Sets the value of the attrname attribute in the
MBO that is bound to the varname variable. The
NOACCESSCHECK flag is set during this update.
Sets the value of the attrname attribute in the
launch point MBO. The NOACCESSCHECK flag is
set during this update.
Sets the value of the attrname attribute to NULL
in the launch point MBO.
Sets the value of the attrname attribute to NULL
in the MBO that is bound to the varname
variable.
Creates an MXException error using the errgrp
group and errkey key.
Makes the attrname attribute mandatory or
optional. You can set bool to 1 for mandatory or 0
for optional.
Makes the attrname attribute read-only or
editable. You can set bool to 1 for read-only or 0
for editable.
Makes the attrname attribute hidden or visible.
You can set bool to 1 for hidden or 0 for visible.
Deletes the MBO that is bound to the varname
variable.
Deletes all MBOs from the MBOSet for the
relation relationship in the launch point MBO.
Deletes all MBOs from the MBOSet for the
relation relationship in the MBO that is bound to
the varname variable.

isnullf(expr)
invokeworkflow(wfname)
maxprop(propname)
maxcond(condname)
tobeadded()
tobesaved()
tobedeleted()
setevalresult(TRUE/FALSE)

setthisattrvalue(value)
scriptvar(varname)

Returns TRUE if the expression evaluates to
NULL.
Invokes the wfname workflow synchronously.
Returns the equivalent string value of the
propname system property.
Evaluates the condname system condition and
returns either TRUE or FALSE.
Returns TRUE if the launch point MBO is marked
for creation.
Returns TRUE if the launch point MBO is marked
for saving.
Returns TRUE if the launch point MBO is marked
for deletion.
Sets the Boolean value of the evalresult implicit
variable in scripts that have a condition launch
point.
Initializes the value of the launch point attribute
in scripts that have an attribute launch point.
Returns the value of the varname implicit
variable. You must specify varname within
quotation marks. This function returns values for
only the following implicit variables:
• app
• action
• interactive
• launchPoint
• mboname
• onadd
• ondelete
• onupdate
• scriptName
• user

MBOSet aggregation functions
Function
countf(relation[, dateattr, duration])

avgf(relation, attrname[, dateattr, duration])

Description
Returns the total number of objects in the
MBOSet for the relation relationship.
You can filter the set of objects that are counted
by specifying a date attribute name and a
duration. An object is included if the dateattr
date is between the current date and a duration
period of time before the current date.
Returns the average value of the attrname
attribute for all objects in the MBOSet for the
relation relationship.

maxf(relation, attrname[, dateattr, duration])

minf(relation, attrname[, dateattr, duration])

You can filter the set of objects that are counted
by specifying a date attribute name and a
duration. An object is included if the dateattr
date is between the current date and a duration
period of time before the current date.
Returns the maximum value of the attrname
attribute for all objects in the MBOSet for the
relation relationship.
You can filter the set of objects that are counted
by specifying a date attribute name and a
duration. An object is included if the dateattr
date is between the current date and a duration
period of time before the current date.
Returns the minimum value of the attrname
attribute for all objects in the MBOSet for the
relation relationship.
You can filter the set of objects that are counted
by specifying a date attribute name and a
duration. An object is included if the dateattr
date is between the current date and a duration
period of time before the current date.

Control flow functions
Function
if(condition, valueiftrue[, valueiffalse])

foreachmbo(relname, scriptname|functionname
[, cond])

foreachtoken(str, splitchar,
scriptname|functionname)
continue()
break()

Description
Returns the valueiftrue value if the condition
condition evaluates to TRUE. You can specify a
value that is returned when the condition
condition evaluates to FALSE by populating the
valueiffalse parameter.
Runs a library script or local function for each
MBO in the MBOSet for the relation relationship.
The number of iterations cannot exceed the
maximum fetch limit that is configured in the
application.
Runs a library script or local function for each
token in the str string. The str string is delimited
into tokens by the splitchar delimiter.
Continues to the next iteration in the loop.
Breaks from the loop.

String manipulation functions
Function
concat(str1, str2)
tolower(str)

Description
Concatenates two strings.
Converts a string to lowercase.

toupper(str)
startswith(str1, str2)
endswith(str1, str2)
substring(str, startofffset[, endoffset])

tokenat(str, splitchar, offset)

Converts a string to uppercase.
Returns TRUE when the str1 string starts with the
str2 string.
Returns TRUE when the str1 string ends with the
str2 string.
Returns a substring of the str string starting at the
startofffset offset. If you do not specify an end
offset in the endoffset position, the function
returns the substring from the startofffset offset
to the end of the string.
Returns the token at the offset offset for the
string str that is delimited by the splitchar
delimiter.

Date manipulation functions
Function
now()
duration(years, months, days, hours, mins, secs)

date(year, mon, day)
datetime(year, mon, day, hh, mm, ss)

Description
Returns the current date and time.
Returns an amount of time specified in years,
months, days, hours, minutes, and seconds. You
can specify a duration that you can add to or
subtract from a date.
Creates a date object for a specific year, month,
and day.
Creates a date object for a specific year, month,
day, and time. You must use two digits to specify
each of the hours, minutes, and seconds
parameters.

Other utility functions
Function
nvl(expr1, expr2)

number(str)
str(num)

Description
Returns the expr1 expression if the expr1
expression is not NULL. If the expr1 expression is
NULL, this function returns the expr2 expression.
Converts a string to a number.
Converts a number to a string.

Boolean primitive functions
Token
TRUE
FALSE

Description
Returns a Boolean TRUE value.
Returns a Boolean FALSE value.

Supported object attribute access notations

You can access MBO attributes directly in your MBR script code by using the following notations:
Notation
relation1$relation2$....relationN$attr

owner1$owner2$....ownerN$attr
modified$attr

prev$attr
internal$attr

Description
Returns the value of an attribute in a related
MBO. You can traverse several relationships to
find the MBO that contains the attribute value
that you want.
Returns the value of an attribute in the launch
point MBO.
Returns TRUE if the attr attribute is modified.
Returns FALSE if the attr attribute is not
modified.
Returns the initial value of an attribute.
Returns the internal value of an attribute that is
bound to a synonym domain.

Script code examples
Example 1: Adding new MBOs to an MBOSet
Use case: When a new asset is created that has an asset type of GASENG, add two meters, O-PRESSUR
and IN-PRESSUR, to the meter MBOSet of the asset.
To create this script:
1.
2.
3.
4.
5.
6.
7.

In the Automation Scripts application, create a script that has an object launch point.
In step 1 of the wizard, specify the launch point name and select the ASSET object.
In the Events section, select the Save radio button.
In the Save section, check the Add check box and select the Before Save radio button.
In the Script section, select the New radio button.
In step 2 of the wizard, specify the script name and set the log level to debug.
In step 3 of the wizard, paste in the following source code and then click Create:

#my first MBR code for asset
setvar("isgaseng",(assettype == "GASENG" && countf("assetmeter")==0))
if(getvar("isgaseng"),invokescript(":createmeters"))

#my MBR function that will add the two meter MBOs to the asset MBO
:createmeters
newmbo("opressure","assetmeter")
setvaluetombo("opressure","metername","O-PRESSUR")
newmbo("inpressure","assetmeter")
setvaluetombo("inpressure","metername","IN-PRESSUR")

Code analysis
This script contains two lines of main code and a function. The script is run when a user saves changes to
an asset record.
In the main code, the following logic is implemented:
1. Create a local variable and set the variable to TRUE if both of the following conditions are true:
• The asset type is GASENG.
• The meter MBOSet of the asset is empty.
2. Evaluate the local variable. If the local variable is TRUE, call a function to add MBOs to the meter
MBOSet of the asset.
In the function, which is called createmeters, the following logic is implemented:
1.
2.
3.
4.

Add an MBO to the meter MBOSet of the asset.
Set the name of the MBO to O-PRESSUR.
Add a second MBO to the meter MBOSet of the asset.
Set the name of the second MBO to IN-PRESSUR.

The script uses two internal variables, assettype and assetmeter, to derive the asset type and the
contents of the meter MBOSet of the asset.
Because this script runs when a user saves their changes, the application transaction framework saves
and commits the newly created MBOs as part of the main transaction.
You can move the createmeters function code into a library script. Remember that library script names
do not contain : characters. For example, you can move the createmeters function code to a library
script that is also called createmeters and then call the library script in the following statement:
if(getvar("isgaseng"),invokescript("createmeters"))

Example 2: Validating attribute values
Use case: If an asset has a type of GASENG, ensure that the purchase price of the asset cannot contain a
value that is greater than 200.
To create this script:
1. In the Automation Scripts application, create a script that has an attribute launch point.
2. In step 1 of the wizard, specify the launch point name, and select the ASSET object and the
PURCHASEPRICE attribute.
3. In the Events section, select the Validate radio button.
4. In the Script section, select the New radio button.
5. In step 2 of the wizard, specify the script name and set the log level to debug.
6. In step 3 of the wizard, paste in the following source code and then click Create:

if(not(isnullf(assettype)) && assettype=="GASENG" && not(isnullf(purchaseprice)) &&
purchaseprice>200,error("asset","toomuchcost"))

Code analysis
This single-line script is run when the purchase price of an asset is modified.
The script creates an error if all of the following conditions are true:
•
•
•
•

The asset type is valid.
The asset has a type of GASENG.
The purchase price of the asset has a value.
The purchase price contains a value that is greater than 200.

In addition to using the internal variables assettype and assetmeter that are used in Example 1, this
script uses the internal variable purchaseprice to derive the purchase price of the asset.
This script also uses a sample error group and error key. When you write script code that creates an
error, ensure that the message group and message key that you use exist in the error message database.

Example 3: Manipulating dates
Use case: If an asset has a type of BUS, perform the following actions:
•
•

Set the end-of-life date to the install date plus one year. If the asset does not have an install
date, set the end-of-life date to the current date plus one year.
Make the priority attribute mandatory.

To create this script:
1. In the Automation Scripts application, create a script that has an attribute launch point.
2. In step 1 of the wizard, specify the launch point name, and select the ASSET object and the
ASSETTYPE attribute.
3. In the Events section, select the Validate radio button.
4. In the Script section, select the New radio button.
5. In step 2 of the wizard, specify the script name and set the log level to debug.
6. In step 3 of the wizard, paste in the following source code and then click Create:
if(assettype=="BUS",setvalue("estendoflife",nvl(installdate,now())+duration(1,0,0,0,0,0)))
if(assettype=="BUS",setrequired("priority",TRUE),setrequired("priority",FALSE))

Code analysis
This script is run when the type of an asset is modified.
In the script, the following logic is implemented:

•

•

If the asset has a type of BUS, set the end-of-life date:
o Evaluate the install date of the asset.
o If the install date is empty, set the end-of-life date to the current date plus one year.
o If the install date is populated, set the end-of-life date to the install date plus one year.
If the asset has a type of BUS, make the priority attribute mandatory. If the asset does not have
a type of BUS, make the priority attribute optional.

For both of these actions, the script checks to see if the asset has a type of BUS, which is inefficient. You
can improve the script by defining a function, moving both actions into the function, and calling the
function if the asset has a type of BUS. The following code demonstrates this improvement:
if(assettype=="BUS",invokescript(":seteol"),setrequired("priority",FALSE))

:seteol
setvalue("estendoflife",nvl(installdate,now())+duration(1,0,0,0,0,0))
setrequired("priority",TRUE)

The script uses the internal variables assettype, estendoflife, and installdate to derive the values of the
type, end-of-life date, and install date of the asset.

Example 4: Looping
Use case: Two types of meter, out-pressure and in-pressure, are defined for an asset. When a reading is
recorded for an asset meter, set the priority of the asset to the numeric difference between the outpressure and in-pressure meter readings.
To create this script:
1.
2.
3.
4.
5.
6.
7.

In the Automation Scripts application, create a script that has an object launch point.
In step 1 of the wizard, specify the launch point name and select the ASSET object.
In the Events section, select the Save radio button.
In the Save section, check the Add check box and select the Before Save radio button.
In the Script section, select the New radio button.
In step 2 of the wizard, specify the script name and set the log level to debug.
In step 3 of the wizard, paste in the following source code and then click Create:

foreachmbo("ACTIVEASSETMETER",":pressurediff")
setvalue("priority",getvar("op")-getvar("ip"))

:pressurediff
if(metername=="O-PRESSUR",setvar("op",number(newreading),TRUE))
if(metername=="IN-PRESSUR",setvar("ip",number(newreading),TRUE))

Code analysis
This script contains two lines of main code and a function. The script is run when a user saves changes to
an asset record.
In the main code, the following logic is implemented:
•
•

For each type of meter that is defined for an asset, call a function that records the meter
reading.
Set the priority of the asset to the numeric difference between the out-pressure and in-pressure
meter readings.

In the function, the following logic is implemented:
•
•

If the function is called for the out-pressure meter, record the out-pressure meter reading in a
variable that has a global scope.
If the function is called for the in-pressure meter, record the in-pressure meter reading in a
variable that has a global scope.

Because the meter readings are assigned to variables that have a global scope, the main code can use
these meter readings to set the priority value on the asset.
The script uses the internal variables priority, metername, and newreading to derive the values of the
priority of the asset and the name and reading of the meter. The main code uses the internal relation
name ACTIVEASSETMETER to determine the meter MBOSet for the asset that the loop iterates through.

Example 5: Setting attributes on duplicate objects
Use case: When a work order is duplicated, copy the original work order number into a custom attribute
in the duplicate work order.
To create this script:
1.
2.
3.
4.

In the Automation Scripts application, create a script that does not have a launch point.
In the wizard, specify a script name of WORKORDER.DUPLICATE.
Specify MBR as the script language.
Paste in the following source code and then click Create:

setvaluetombo("dupmbo","copiedfrom", wonum)

Code analysis
The script name ensures that this script is run when a work order is duplicated.
The script copies the original work order number into a custom attribute in the duplicate work order.

The script uses the internal variables wonum, dupmbo, and copiedfrom to derive the values of the
original work order number, the duplicate work order number, and the custom attribute on the
duplicate work order.

